
You must be able to

write a count-controlled loop, including one with STEP
write a pre-condition loop and a post-condition loop, and change one kind of loop
into another
use the standard loop patterns: a running total, a count, the largest or smallest
value, a validation loop and a rogue value 终止值
use an IF statement with an ELSE clause, nested 嵌套 IF statements or a CASE
structure inside a loop
write nested loops
justify why one loop structure is better suited to a problem than the others
dry-run a loop with a trace table 跟踪表



1 The method

Method — Three loops, one task
Each of these outputs the numbers 1 to 5.

FOR Count ← 1 TO 5
OUTPUT Count

NEXT Count

Count ← 1
WHILE Count <= 5

OUTPUT Count
Count ← Count + 1

ENDWHILE

Method — Three loops, one task

Count ← 1
REPEAT

OUTPUT Count
Count ← Count + 1

UNTIL Count > 5

Method — Three loops, one task

A FOR loop sets, tests and increases its counter for you. The other two loops need
all three steps written out: set the counter before the loop, change it inside the
loop, and test it.
WHILE repeats while its condition is true; REPEAT repeats until its condition is
true. So the two conditions are opposites: Count <= 5 and Count > 5.
STEP changes the counter by a different amount: FOR Count ← 10 TO 0 STEP
-2 gives 10, 8, 6, 4, 2, 0.

Method — Loop patterns: a total and a count

Total ← 0
BelowZero ← 0
FOR Day ← 1 TO 30

INPUT Temp
Total ← Total + Temp
IF Temp < 0 THEN

BelowZero ← BelowZero + 1
ENDIF

NEXT Day
OUTPUT BelowZero
OUTPUT Total / 30

Method — Loop patterns: a total and a count

Running total: set it to 0 before the loop, and add to it inside the loop.
Count: set it to 0, and add 1 inside an IF.
Declare Day and BelowZero as INTEGER, and Temp and Total as REAL.

Method — Loop pattern: the largest value

FOR Day ← 1 TO 30
INPUT Temp
IF Day = 1 THEN

Highest ← Temp
ELSE

IF Temp > Highest THEN
Highest ← Temp

ENDIF
ENDIF

NEXT Day
OUTPUT Highest

Method — Loop pattern: the largest value

Start with the first value, not with 0, because every temperature could be
below zero. The ELSE clause holds a nested IF that only runs from day 2
onwards.

Method — A validation loop

REPEAT
OUTPUT “Enter a mark from 0 to 100”
INPUT Mark

UNTIL Mark >= 0 AND Mark <= 100

The mark must be input at least once before it can be checked, so a post-condition loop
fits.
With a pre-condition loop, input the mark once before the loop and again as the last
statement inside it, and loop WHILE Mark < 0 OR Mark > 100.



11.2.1 Writing, choosing and tracing loops 11 / 49

Method — A rogue value

Input prices one at a time until -1 is entered, then output the total price. The value -1 is a
rogue value: it ends the loop, and it must not be added to the total.

Total ← 0
INPUT Price
WHILE Price <> -1

Total ← Total + Price
INPUT Price

ENDWHILE
OUTPUT Total

Method — A rogue value

A pre-condition loop is used, because the very first input could be -1, and
then nothing should be added.
Inside the loop, process the value, then input the next one, so every
value is tested before it is used.

Method — Nested loops

INPUT Rows
FOR Row ← 1 TO Rows

Line ← “”
FOR Star ← 1 TO Row

Line ← Line & “*”
NEXT Star
OUTPUT Line

NEXT Row

Method — Nested loops

The inner loop runs completely for each pass of the outer loop.
OUTPUT starts a new line each time, so the stars for one row are joined into
Line first.
When Rows is 3, the statement inside the inner loop runs 1 + 2 + 3 = 6
times.



11.2.1 Writing, choosing and tracing loops 15 / 49

Method — Choosing and justifying a loop

the problem best loop justification
process exactly 30
readings

count-
controlled

the number of repetitions is known
before the loop starts

read values until a rogue
value, which could come
first

pre-condition the condition is tested before the first
pass, so the loop can run zero times

input a password until it
is correct

post-condition the body must run at least once,
because the value to test is input
inside the loop

In an exam answer, name the loop and give the reason in terms of the problem.

Method — Dry-running a loop

Trace this algorithm for the inputs 4, 7, 2 and 0.

Count ← 0
Big ← 0
INPUT N
WHILE N <> 0

Count ← Count + 1
IF N > Big THEN

Big ← N
ENDIF
INPUT N

ENDWHILE
OUTPUT Count, “ ”, Big

Method — Dry-running a loop

N Count Big OUTPUT
0 0

4 1 4
7 2 7
2 3
0 3 7

Method — Dry-running a loop

Write a value only when a variable changes, and move down a
row as the algorithm runs.



2 Practice

Practice 2.1 • 3 marks

Rewrite this count-controlled loop as a pre-condition loop that gives the same output.

FOR Number ← 3 TO 15 STEP 3
OUTPUT Number

NEXT Number



11.2.1 Writing, choosing and tracing loops 20 / 49

Solution 2.1

Method: Three loops, one task — p.4

Worked steps

set Number to 3 before the loop B1
WHILE Number <= 15 … ENDWHILE B1
output, then add 3 inside the loop B1

Number ← 3
WHILE Number <= 15

OUTPUT Number
Number ← Number + 3

ENDWHILE

Practice 2.2 • 3 marks

Write pseudocode that inputs a person’s age, and repeats the input
until the age is from 0 to 120. Use a post-condition loop.

Solution 2.2

Worked steps

REPEAT … UNTIL structure B1

INPUT Age inside the loop B1

the condition Age >= 0 AND Age <= 120 B1

REPEAT
INPUT Age

UNTIL Age >= 0 AND Age <= 120

Practice 2.3 • 4 marks

Write pseudocode that inputs the heights of 25 students, one at a
time, and outputs the smallest height.

Solution 2.3

Method: Dry-running a loop — p.17

Worked steps

input the first height and store it as the smallest B1

a loop that runs for the other 24 heights B1

compare each height with the smallest, and replace it if smaller B1

output the smallest after the loop B1

Solution 2.3

DECLARE Count : INTEGER
DECLARE Height, Smallest : REAL
INPUT Height
Smallest ← Height
FOR Count ← 2 TO 25

INPUT Height
IF Height < Smallest THEN

Smallest ← Height
ENDIF

NEXT Count
OUTPUT Smallest

Practice 2.4 • 2 marks

A program reads words from the user until the user types “END”.
The user may type “END” as the first word. State the most suitable
loop structure, and justify your choice.

Solution 2.4

Worked steps

a pre-condition loop B1

the condition is tested before any word is processed, so the loop can run zero
times when “END” is typed first B1

Practice 2.5 • 3 marks

Complete the trace table for this algorithm.

Total ← 0
FOR Outer ← 1 TO 3

FOR Inner ← 1 TO Outer
Total ← Total + Inner

NEXT Inner
NEXT Outer
OUTPUT Total

Practice 2.5 • 3 marks

Outer Inner Total OUTPUT
0

1 1 1

Solution 2.5

Method: Nested loops — p.14

Worked steps

Outer Inner Total OUTPUT

0
1 1 1
2 1 2

2 4
3 1 5

2 7
3 10

10

Solution 2.5

Outer and Inner columns correct B1 ; Total column correct B1 ; output 10
B1



3 Exam-style

Exam-style 3.1 • 7 marks

A weather station records the rainfall, in mm, for each of the 30 days of a month. An
algorithm will:

input the rainfall for each day, one value at a time
count the days with no rain
find the total rainfall for the month
find the day with the most rain; if two days have the same rainfall, keep the earlier
day
output the count, the total and the day number.

Write pseudocode for the algorithm. Declare any variables you use.

Solution 3.1

Method: Loop patterns: a total and a count — p.7

Worked steps

declarations B1 ; a count-controlled loop from 1 to 30 B1 ; input inside the loop B1

count the days with rainfall 0 B1 ; running total B1

keep the largest rainfall and its day number, replacing only when greater B1

output all three after the loop B1

Solution 3.1

DECLARE Day, DryDays, WettestDay : INTEGER
DECLARE Rain, Total, MostRain : REAL
DryDays ← 0
Total ← 0
MostRain ← -1
WettestDay ← 0
FOR Day ← 1 TO 30

INPUT Rain
IF Rain = 0 THEN

DryDays ← DryDays + 1
ENDIF
Total ← Total + Rain
IF Rain > MostRain THEN

MostRain ← Rain
WettestDay ← Day

ENDIF
NEXT Day
OUTPUT DryDays
OUTPUT Total
OUTPUT WettestDay

Solution 3.1

MostRain starts at -1, below any real rainfall, so day 1 is always stored first;
> rather than >= keeps the earlier day

Exam-style 3.2 • 8 marks

A calculator program repeats a menu until the user chooses to stop.

choice action
1 input a number and add it to Total
2 input a number and multiply Total by it
3 output Total
4 output “Stopped”, and the program ends

Exam-style 3.2 (a) • 6 marks

Write pseudocode for the program, using a CASE structure inside a
loop. Total starts at 1, and any other choice outputs “Invalid”.

Solution 3.2 (a)

Method: A rogue value — p.12

Worked steps

a loop that ends when the choice is 4 B1 ; input the choice inside the loop B1
CASE OF … ENDCASE B1 ; choices 1 and 2 B1 ; choice 3 B1 ; choices 4 and OTHERWISE B1

DECLARE Choice : INTEGER
DECLARE Total, Number : REAL
Total ← 1
REPEAT

INPUT Choice
IF Choice = 1 OR Choice = 2 THEN

INPUT Number
ENDIF
CASE OF Choice

1 : Total ← Total + Number
2 : Total ← Total * Number
3 : OUTPUT Total
4 : OUTPUT “Stopped”
OTHERWISE : OUTPUT “Invalid”

ENDCASE
UNTIL Choice = 4

Solution 3.2 (a)

(inputting Number inside the clauses for 1 and 2 of the CASE
structure is also correct)

Exam-style 3.2 (b) • 2 marks

Justify the loop structure you used.

Solution 3.2 (b)

Worked steps

a post-condition loop B1

the choice must be input at least once before it can be tested B1

Exam-style 3.3 • 9 marks
Study this pseudocode.

Count ← 0
Total ← 0
INPUT Value
WHILE Value <> 999

IF Value > 0 THEN
Total ← Total + Value
Count ← Count + 1

ENDIF
INPUT Value

ENDWHILE
IF Count > 0 THEN

OUTPUT Total / Count
ELSE

OUTPUT “No data”
ENDIF

Exam-style 3.3 (a) • 4 marks

Complete the trace table for the inputs 6, -2, 9, 0 and 999.

Value Count Total OUTPUT
0 0

Solution 3.3 (a)

Method: Dry-running a loop — p.17

Worked steps

Value Count Total OUTPUT

0 0
6 1 6
-2
9 2 15
0
999 7.5

Solution 3.3 (a)

Value column B1 ; Count column B1 ; Total column B1 ; output 7.5 B1

Exam-style 3.3 (b) • 2 marks

State the purpose of the algorithm.

Solution 3.3 (b)

Worked steps

it outputs the mean (average) of the positive values input B1

it stops at the rogue value 999, and outputs “No data” if there were no
positive values B1

Exam-style 3.3 (c) • 3 marks

The algorithm is rewritten with a post-condition loop, and INPUT
Value becomes the first statement inside the loop. Explain what
goes wrong if the first value input is 999, and how to correct it.

Solution 3.3 (c)

Worked steps

the test comes at the end of the loop, so 999 is processed before the loop stops B1

999 is positive, so it is added to Total and counted, and the output is 999
instead of “No data” B1

correct it by changing the condition to Value > 0 AND Value <> 999 B1

